

Even Wins

This is a game between you and the computer. To play, an odd number of objects (marbles, chips, matches) are placed in a row. You take turns with the computer picking up between one and four objects each turn. The game ends when there are no objects left, and the winner is the one with an even number of objects picked up.

Two versions of this game are included. While to the player they appear similar, the programming approach is quite different. EVEN WINS, the first version, is deterministic — i.e., the computer plays by fixed, good rules and is impossible to beat if you don't know how to play the game. It always starts with 27 objects, although you may change this in Lines 250, and 1060.

The second version, GAME OF EVEN WINS, is much more interesting because the computer starts out only knowing the rules of the game. Using simple techniques of artificial intelligence (cybernetics), the computer gradually learns to play from its mistakes until it plays a very good game. After 20 games, the computer is a challenge to beat. Variation in the human's style of play seems to make the computer learn more quickly. If you plot the learning curve of this program, it closely resembles classical human learning curves from psychological experiments.

Eric Peters at DEC wrote the GAME OF EVEN WINS. The original author of EVEN WINS is unknown.

EVEN WINS
CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

THIS IS A TWO PERSON GAME CALLED 'EVEN WINS.'
TO PLAY THE GAME, THE PLAYERS NEED 27 MARBLES OR
OTHER OBJECTS ON A TABLE.

THE 2 PLAYERS ALTERNATE TURNS, WITH EACH PLAYER
REMOVING FROM 1 TO 4 MARBLES ON EACH MOVE. THE GAME
ENDS WHEN THERE ARE NO MARBLES LEFT, AND THE WINNER
IS THE ONE WITH AN EVEN NUMBER OF MARBLES.

THE ONLY RULES ARE THAT (1) YOU MUST ALTERNATE TURNS,
(2) YOU MUST TAKE BETWEEN 1 AND 4 MARBLES EACH TURN,
AND (3) YOU CANNOT SKIP A TURN.

TYPE A 1 IF YOU WANT TO GO FIRST, AND TYPE
A 0 IF YOU WANT ME TO GO FIRST.

? 0

TOTAL= 27

I PICK UP 2 MARBLES. •

TOTAL= 25

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 2

? 2

TOTAL= 23

YOUR TOTAL IS 2

I PICK UP 4 MARBLES.

TOTAL= 19

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 6

? 4

TOTAL= 15

YOUR TOTAL IS 6

I PICK UP 2 MARBLES.

TOTAL= 13

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 8

? 1

TOTAL= 12

YOUR TOTAL IS 7

I PICK UP 1 MARBLES.

TOTAL= 11

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 9

? 3

TOTAL= 8

YOUR TOTAL IS 10

I PICK UP 1 MARBLES.

TOTAL= 7

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 10

? 1

TOTAL= 6

YOUR TOTAL IS 11

I PICK UP 1 MARBLES.

TOTAL= 5

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 11

? 1

TOTAL= 4

YOUR TOTAL IS 12

I PICK UP 3 MARBLES.

TOTAL= 1

AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS 14

? 1

THAT IS ALL OF THE MARBLES.

MY TOTAL IS 14 YOUR TOTAL IS 13

I WON. DO YOU WANT TO PLAY
AGAIN? TYPE 1 FOR YES AND 0 FOR NO.

```

1 PRINT TAB(31);"EVEN WINS"
2 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
3 PRINT:PRINT:PRINT
4 Y1=0
10 M1=0
20 DIM M(20),Y(20)
30 PRINT "      THIS IS A TWO PERSON GAME CALLED 'EVEN WINS.'"
40 PRINT "TO PLAY THE GAME, THE PLAYERS NEED 27 MARBLES OR"
50 PRINT "OTHER OBJECTS ON A TABLE."
60 PRINT
70 PRINT
80 PRINT "      THE 2 PLAYERS ALTERNATE TURNS, WITH EACH PLAYER"
90 PRINT "REMOVING FROM 1 TO 4 MARBLES ON EACH MOVE. THE GAME"
100 PRINT "ENDS WHEN THERE ARE NO MARBLES LEFT, AND THE WINNER"
110 PRINT "IS THE ONE WITH AN EVEN NUMBER OF MARBLES."
120 PRINT
130 PRINT
140 PRINT "      THE ONLY RULES ARE THAT (1) YOU MUST ALTERNATE TURNS,"
150 PRINT "(2) YOU MUST TAKE BETWEEN 1 AND 4 MARBLES EACH TURN,"
160 PRINT "AND (3) YOU CANNOT SKIP A TURN."
170 PRINT
180 PRINT
190 PRINT
200 PRINT "      TYPE A 1 IF YOU WANT TO GO FIRST, AND TYPE"
210 PRINT "A 0 IF YOU WANT ME TO GO FIRST."
220 INPUT C
230 IF C=0 THEN 250
240 GOTO 1060
250 T=27
260 M=2
270 PRINT "TOTAL=";T
280 M1=M1+M
290 T=T-M
300 PRINT "I PICK UP";M;"MARBLES."
310 IF T=0 THEN 880
320 PRINT "TOTAL=";T
330 PRINT
340 PRINT "      AND WHAT IS YOUR NEXT MOVE, MY TOTAL IS";M1
350 INPUT Y
360 PRINT
370 IF Y<1 THEN 1160
380 IF Y>4 THEN 1160
390 IF Y<=T THEN 430
400 PRINT "      YOU HAVE TRIED TO TAKE MORE MARBLES THAN THERE ARE"
410 PRINT "LEFT. TRY AGAIN."
420 GOTO 350
430 Y1=Y1+Y
440 T=T-Y
450 IF T=0 THEN 880
460 PRINT "TOTAL=";T
470 PRINT
480 PRINT "      YOUR TOTAL IS";Y1
490 IF T<.5 THEN 880
500 R=T-6*INT(T/6)
510 IF INT(Y1/2)=Y1/2 THEN 700
520 IF T<4.2 THEN 580
530 IF R>3.4 THEN 620
540 M=R+1
550 M1=M1+M
560 T=T-M
570 GOTO 300
580 M=T
590 T=T-M

```

```

600 GOTO 830
610 REM      250 IS WHERE I WIN.
620 IF R<4.7 THEN 660
630 IF R>3.5 THEN 660
640 M=1
650 GOTO 670
660 M=4
670 T=T-M
680 M1=M1+M
690 GOTO 300
700 REM      I AM READY TO ENCODE THE STRAT FOR WHEN OPP TOT IS EVEN
710 IF R<1.5 THEN 1020
720 IF R>5.3 THEN 1020
730 M=R-1
740 M1=M1+M
750 T=T-M
760 IF T<.2 THEN 790
770 REM      IS M ZERO HERE
780 GOTO 300
790 REM      IS = ZERO HERE
800 PRINT "I PICK UP";M;"MARBLES."
810 PRINT
820 GOTO 880
830 REM      THIS IS WHERE I WIN
840 PRINT "I PICK UP";M;"MARBLES."
850 PRINT
860 PRINT "TOTAL = 0"
870 M1=M1+M
880 PRINT "      THAT IS ALL OF THE MARBLES."
890 PRINT
900 PRINT "      MY TOTAL IS";M1;"      YOUR TOTAL IS";Y1
910 PRINT
920 IF INT(M1/2)=M1/2 THEN 950
930 PRINT "      YOU WON. DO YOU WANT TO PLAY"
940 GOTO 960
950 PRINT "      I WON. DO YOU WANT TO PLAY"
960 PRINT "AGAIN? TYPE 1 FOR YES AND 0 FOR NO."
970 INPUT A1
980 IF A1=0 THEN 1030
990 M1=0
1000 Y1=0
1010 GOTO 200
1020 GOTO 640
1030 PRINT
1040 PRINT "OK. SEE YOU LATER."
1050 GOTO 1230
1060 T=27
1070 PRINT
1080 PRINT
1090 PRINT
1100 PRINT "TOTAL=";T
1110 PRINT
1120 PRINT
1130 PRINT "      WHAT IS YOUR FIRST MOVE?"
1140 INPUT Y
1150 GOTO 360
1160 PRINT
1170 PRINT "THE NUMBER OF MARBLES YOU TAKE MUST BE A POSITIVE"
1180 PRINT "INTEGER BETWEEN 1 AND 4."
1190 PRINT
1200 PRINT "      WHAT IS YOUR NEXT MOVE?"
1210 PRINT
1220 GOTO 350
1230 END

```

GAME OF EVEN WINS
CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

DO YOU WANT INSTRUCTIONS (YES OR NO)? YES

THE GAME IS PLAYED AS FOLLOWS:
AT THE BEGINNING OF THE GAME, A RANDOM NUMBER OF CHIPS ARE
PLACED ON THE BOARD. THE NUMBER OF CHIPS ALWAYS STARTS
AS AN ODD NUMBER. ON EACH TURN, A PLAYER MUST TAKE ONE,
TWO, THREE, OR FOUR CHIPS. THE WINNER IS THE PLAYER WHO
FINISHES WITH A TOTAL NUMBER OF CHIPS THAT IS EVEN.
THE COMPUTER STARTS OUT KNOWING ONLY THE RULES OF THE
GAME. IT GRADUALLY LEARNS TO PLAY WELL. IT SHOULD BE
DIFFICULT TO BEAT THE COMPUTER AFTER TWENTY GAMES IN A ROW.
TRY IT!!!!

TO QUIT AT ANY TIME, TYPE A '0' AS YOUR MOVE.

THERE ARE 21 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 17 ... YOUR MOVE? 4
THERE ARE 13 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 9 ... YOUR MOVE? 2
THERE ARE 7 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 3 ... YOUR MOVE? 1
THERE ARE 2 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS.
GAME OVER ... I WIN!!!

THERE ARE 19 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 15 ... YOUR MOVE? 4
THERE ARE 11 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 7 ... YOUR MOVE? 2
THERE ARE 5 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 1 ... YOUR MOVE? 1
GAME OVER ... I WIN!!!

THERE ARE 9 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 5 ... YOUR MOVE? 2
THERE ARE 3 CHIPS ON THE BOARD.
COMPUTER TAKES 3 CHIPS.
GAME OVER ... YOU WIN!!!

THERE ARE 21 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 19 ... YOUR MOVE? 4
THERE ARE 17 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 13 ... YOUR MOVE? 1
THERE ARE 12 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 8 ... YOUR MOVE? 3
THERE ARE 5 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 1 ... YOUR MOVE? 1
GAME OVER ... I WIN!!!

THERE ARE 9 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 7 ... YOUR MOVE? 4
THERE ARE 3 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 1 ... YOUR MOVE? 1
GAME OVER ... I WIN!!!

THERE ARE 21 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 19 ... YOUR MOVE? 1
THERE ARE 18 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 14 ... YOUR MOVE? 1
THERE ARE 13 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 9 ... YOUR MOVE? 1
THERE ARE 8 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 6 ... YOUR MOVE? 1
THERE ARE 5 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 1 ... YOUR MOVE? 1
GAME OVER ... I WIN!!!

THERE ARE 9 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 7 ... YOUR MOVE? 4
THERE ARE 3 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 1 ... YOUR MOVE? 1
GAME OVER ... I WIN!!!

THERE ARE 21 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 19 ... YOUR MOVE? 4
THERE ARE 15 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS LEAVING 13 ... YOUR MOVE? 3
THERE ARE 10 CHIPS ON THE BOARD.
COMPUTER TAKES 4 CHIPS LEAVING 6 ... YOUR MOVE? 4
THERE ARE 2 CHIPS ON THE BOARD.
COMPUTER TAKES 2 CHIPS.
GAME OVER ... I WIN!!!

```
1 PRINT TAB(28);"GAME OF EVEN WINS"
2 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
3 PRINT:PRINT:PRINT
4 INPUT "DO YOU WANT INSTRUCTIONS (YES OR NO)";A$
5 IF A$="NO" THEN 20
6 PRINT: PRINT "THE GAME IS PLAYED AS FOLLOWS:"
7 PRINT "AT THE BEGINNING OF THE GAME, A RANDOM NUMBER OF CHIPS ARE"
8 PRINT "PLACED ON THE BOARD. THE NUMBER OF CHIPS ALWAYS STARTS"
9 PRINT "AS AN ODD NUMBER. ON EACH TURN, A PLAYER MUST TAKE ONE,"
10 PRINT "TWO, THREE, OR FOUR CHIPS. THE WINNER IS THE PLAYER WHO"
11 PRINT "FINISHES WITH A TOTAL NUMBER OF CHIPS THAT IS EVEN."
12 PRINT "THE COMPUTER STARTS OUT KNOWING ONLY THE RULES OF THE"
13 PRINT "GAME. IT GRADUALLY LEARNS TO PLAY WELL. IT SHOULD BE"
14 PRINT "DIFFICULT TO BEAT THE COMPUTER AFTER TWENTY GAMES IN A ROW."
15 PRINT "TRY IT!!!!": PRINT
16 PRINT "TO QUIT AT ANY TIME, TYPE A '0' AS YOUR MOVE.": PRINT
20 DIM R(1,5)
25 L=0: B=0
30 FOR I=0 TO 5
40 R(1,I)=4
50 R(0,I)=4
60 NEXT I
70 A=0: B=0
90 P=INT((13*RND(1)+9)/2)*2+1
100 IF P=1 THEN 530
110 PRINT "THERE ARE";P;"CHIPS ON THE BOARD."
120 E1=E
130 L1=L
140 E=(A/2-INT(A/2))*2
150 L=INT((P/6-INT(P/6))*6+.5)
160 IF R(E,L)>P THEN 320
170 M=R(E,L)
180 IF M<=0 THEN 370
190 P=P-M
200 IF M=1 THEN 510
210 PRINT "COMPUTER TAKES";M;"CHIPS LEAVING";P;"... YOUR MOVE";
220 B=B+M
230 INPUT M
240 M=INT(M)
250 IF M<1 THEN 450
260 IF M>4 THEN 460
270 IF M>P THEN 460
280 IF M=P THEN 360
290 P=P-M
300 A=A+M
310 GOTO 100
320 IF P=1 THEN 550
330 PRINT "COMPUTER TAKES";P;"CHIPS."
340 R(E,L)=P
350 B=B+P
360 IF B/2=INT(B/2) THEN 420
370 PRINT "GAME OVER ... YOU WIN!!!": PRINT
390 IF R(E,L)=1 THEN 480
400 R(E,L)=R(E,L)-1
410 GOTO 70
420 PRINT "GAME OVER ... I WIN!!!": PRINT
430 GOTO 70
450 IF M=0 THEN 570
460 PRINT M;"IS AN ILLEGAL MOVE ... YOUR MOVE";
470 GOTO 230
480 IF R(E1,L1)=1 THEN 70
490 R(E1,L1)=R(E1,L1)-1
500 GOTO 70
510 PRINT "COMPUTER TAKES 1 CHIP LEAVING";P;"... YOUR MOVE";
520 GOTO 220
530 PRINT "THERE IS 1 CHIP ON THE BOARD."
540 GOTO 120
550 PRINT "COMPUTER TAKES 1 CHIP."
560 GOTO 340
570 END
```