

ICBM

Your radar station picks up an enemy ICBM heading your way, telling you its coordinates (in miles north and miles east of your location). You launch a surface-to-air missile (SAM) to intercept it.

Your only control over the SAM is that you can aim it in any direction, both at launch, and in mid-air. Using the coordinates of the ICBM as a guide, you INPUT the direction (measured CCW from North) in which you want the SAM to travel.

At the next radar scan one minute later, you are given the new coordinates of the ICBM, the coordinates of your SAM, and the distance between the two. You can now make corrections in the course of your SAM by entering a new direction.

You have no control over the altitude of your SAM, as it is assumed that it will seek the same altitude as the ICBM.

As the two missiles draw closer, you make adjustments in the direction of the SAM so as to intercept the ICBM. It's not easy to hit, because the ICBM is programmed to make evasive maneuvers, by taking random deviations from the straight line course to your location. Also, its speed is not known, although it does not vary after being randomly selected at the start of the run.

You can destroy the ICBM by coming within 5 miles of it, at which time your SAM's heat-seeking sensors will come into action and direct it to its target. If you overshoot

the ICBM it's possible to turn the SAM around and chase the ICBM back towards your location. But be careful; you may get both missiles in your lap.

There is also some element of chance involved, as several accidents have been programmed to occur randomly. These can work for you or against you.

Some ways to improve and expand the program are:

1. Operator control over SAM speed: In the present version the speed of the SAM is randomly selected by the computer at the start of the run, and remains constant thereafter. This often results in overshooting the ICBM. Modify the program so that you can input a new speed (within limits) at the same time you input the new direction.

2. Three dimensional version: Have the computer print the *altitude* of the ICBM, as well as its coordinates. The operator will then have to INPUT the angle his SAM is to make with the horizontal, when entering the other quantities.

3. Extend to all Quadrants. In the present version, the ICBM approaches only from the Northeast. You can expand this to include approach from any compass direction.

This game is derived from a program submitted by Chris Falco. The writeup is by Paul Calter and originally appeared in *Creative Computing*, May/June 1975.

ICBM CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

-----MISSLE-----		-----SAM-----		-----
MILES NORTH	MILES EAST	MILES NORTH	MILES EAST	HEADING ?
201	975	0	0	? 15
ICBM & SAM NOW 920 MILES APART				
198	920	48	12	? 85
ICBM & SAM NOW 822 MILES APART				
200	870	52	61	? 75
ICBM & SAM NOW 715 MILES APART				
196	812	64	109	? 80
ICBM & SAM NOW 613 MILES APART				
180	762	72	158	? 85
ICBM & SAM NOW 511 MILES APART				
169	710	76	207	? 80
ICBM & SAM NOW 395 MILES APART				
157	645	84	256	? 70
ICBM & SAM NOW 298 MILES APART				
157	595	101	302	? 75
ICBM & SAM NOW 186 MILES APART				
140	535	113	350	? 80
ICBM & SAM NOW 84 MILES APART				
126	483	121	399	? 90
ICBM & SAM NOW 27 MILES APART				
113	423	121	449	? 95
ICBM & SAM NOW 137 MILES APART				
113	361	116	498	? -90
ICBM & SAM NOW 153 MILES APART				
110	295	116	448	? -95
TOO BAD. YOUR SAM FELL TO THE GROUND!				
DO YOU WANT TO PLAY MORE? (Y OR N)? N				

```

10 PRINT TAB(26); "ICBM"
20 PRINT TAB(20); "CREATIVE COMPUTING"
30 PRINT TAB(18); "MORRISTOWN, NEW JERSEY"
40 PRINT:PRINT:PRINT
110 X1=0:Y1=0
120 X=INT(RND(1)*800)+200:Y=INT(RND(1)*800)+200
130 S=INT(RND(1)*20+50):S1=INT(RND(1)*20+50)

```

```

170 PRINT "-----MISSLE-----";
175 PRINT "-----SAM-----";
180 PRINT "MILES", "MILES", "MILES", "MILES", "HEADING"
190 PRINT "NORTH", "EAST", "NORTH", "EAST", "?"
200 PRINT "-----";
205 PRINT "-----";
210 FOR N=1 TO 50
220 PRINT Y,X,Y1,X1,
230 IF X=0 THEN 550
240 INPUT T1
250 T1=T1/57.296
260 H=INT(RND(1)*200+1)
270 IF H>4 THEN 290
280 ON H GOTO 470,490,510,530
290 X1=INT(X1+S1*SIN(T1)):Y1=INT(Y1+S1*COS(T1))
310 IF SQR(X^2+Y^2)>S THEN 350
320 X=0:Y=0
340 GOTO 430
350 B=SQR(X^2+Y^2)/1000
360 T=ATN(Y/X)
370 X=INT(X-S*COS(T)+RND(1)*20+R)
380 Y=INT(Y-S*SIN(T)+RND(1)*20+R)
390 D=SQR((X-X1)^2+(Y-Y1)^2)
400 IF D<=5 THEN 440
410 D=INT(D)
420 PRINT "ICBM & SAM NOW"; D; "MILES APART"
430 NEXT N
440 PRINT "CONGRATULATIONS! YOUR SAM CAME WITHIN";D;"MILES OF"
450 PRINT "THE ICBM AND DESTROYED IT!"
460 GOTO 560
470 PRINT "TOO BAD. YOUR SAM FELL TO THE GROUND!"
480 GOTO 560
490 PRINT "YOUR SAM EXPLODED IN MIDAIR!"
500 GOTO 560
510 PRINT "GOOD LUCK-THE ICBM EXPLODED HARMLESSLY IN MIDAIR!"
520 GOTO 560
530 PRINT "GOOD LUCK-THE ICBM TURNED OUT TO BE A FRIENDLY AIRCRAFT!"
540 GOTO 560
550 PRINT "TOO BAD!"
555 PRINT "THE ICBM JUST HIT YOUR LOCATION!!"
560 PRINT "DO YOU WANT TO PLAY MORE? (Y OR N)";
570 INPUT A$
580 IF A$="Y" THEN 130
590 END

```