

# Depth Charge

In this program you are captain of the destroyer USS Computer. An enemy submarine has been causing trouble and your mission is to destroy it. You may select the size of the "cube" of water you wish to search in. The computer then determines how many depth charges you get to destroy the submarine.

Each depth charge is exploded by you specifying a trio of numbers; the first two are the surface coordinates (X,Y), the third is the depth. After each depth charge, your sonar observer will tell you where the explosion was relative to the submarine.

Dana Nofle wrote this program while a student at Acton High School, Acton, Massachusetts.

DEPTH CHARGE  
CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

DEPTH CHARGE GAME

DIMENSION OF SEARCH AREA? 50

YOU ARE THE CAPTAIN OF THE DESTROYER USS COMPUTER  
AN ENEMY SUB HAS BEEN CAUSING YOU TROUBLE. YOUR  
MISSION IS TO DESTROY IT. YOU HAVE 6 SHOTS.  
SPECIFY DEPTH CHARGE EXPLOSION POINT WITH A  
TRIO OF NUMBERS -- THE FIRST TWO ARE THE  
SURFACE COORDINATES; THE THIRD IS THE DEPTH.

GOOD LUCK !

TRIAL # 1 ? 25,25,25  
SONAR REPORTS SHOT WAS SOUTHEAST AND TOO LOW.

TRIAL # 2 ? 12,35,12  
SONAR REPORTS SHOT WAS SOUTHWEST AND TOO LOW.

TRIAL # 3 ? 18,43,5  
SONAR REPORTS SHOT WAS NORTHEAST AND TOO HIGH.

TRIAL # 4 ? 15,39,8  
SONAR REPORTS SHOT WAS EAST AND TOO LOW.

TRIAL # 5 ? 14,39,6

B O O M ! ! YOU FOUND IT IN 5 TRIES!

ANOTHER GAME (Y OR N)? N  
OK. HOPE YOU ENJOYED YOURSELF.

DEPTH CHARGE GAME

DIMENSION OF SEARCH AREA? 10

YOU ARE THE CAPTAIN OF THE DESTROYER USS COMPUTER  
AN ENEMY SUB HAS BEEN CAUSING YOU TROUBLE. YOUR  
MISSION IS TO DESTROY IT. YOU HAVE 4 SHOTS.  
SPECIFY DEPTH CHARGE EXPLOSION POINT WITH A  
TRIO OF NUMBERS -- THE FIRST TWO ARE THE  
SURFACE COORDINATES; THE THIRD IS THE DEPTH.

GOOD LUCK !

TRIAL # 1 ? 5,5,5  
SONAR REPORTS SHOT WAS NORTH AND TOO HIGH.

TRIAL # 2 ? 5,2,7  
SONAR REPORTS SHOT WAS SOUTH AND TOO HIGH.

TRIAL # 3 ? 5,3,9  
SONAR REPORTS SHOT WAS SOUTH AND TOO LOW.

TRIAL # 4 ? 5,4,8

B O O M ! ! YOU FOUND IT IN 4 TRIES!

```

2 PRINT TAB(30);"DEPTH CHARGE"
4 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
6 PRINT: PRINT: PRINT
10 PRINT "DEPTH CHARGE GAME": PRINT
20 INPUT "DIMENSION OF SEARCH AREA";G: PRINT
30 N=INT((LOG(G)/LOG(2)))+1
40 PRINT "YOU ARE THE CAPTAIN OF THE DESTROYER USS COMPUTER"
50 PRINT "AN ENEMY SUB HAS BEEN CAUSING YOU TROUBLE. YOUR"
60 PRINT "MISSION IS TO DESTROY IT. YOU HAVE";N;"SHOTS."
70 PRINT "SPECIFY DEPTH CHARGE EXPLOSION POINT WITH A"
80 PRINT "TRIO OF NUMBERS -- THE FIRST TWO ARE THE"
90 PRINT "SURFACE COORDINATES; THE THIRD IS THE DEPTH."
100 PRINT: PRINT "GOOD LUCK !": PRINT
110 A=INT(G*RND(1)): B=INT(G*RND(1)): C=INT(G*RND(1))
120 FOR D=1 TO N: PRINT: PRINT "TRIAL #";D: INPUT X,Y,Z
130 IF ABS(X-A)+ABS(Y-B)+ABS(Z-C)=0 THEN 300
140 GOSUB 500: PRINT: NEXT D
200 PRINT: PRINT "YOU HAVE BEEN TORPEDDED! ABANDON SHIP!"
210 PRINT "THE SUBMARINE WAS AT";A;"",B;"",C: GOTO 400
300 PRINT: PRINT "B O O M ! ! YOU FOUND IT IN";D;"TRIES!"
400 PRINT: PRINT: INPUT "ANOTHER GAME (Y OR N)";A$
410 IF A$="Y" THEN 100
420 PRINT "OK. HOPE YOU ENJOYED YOURSELF.": GOTO 600
500 PRINT "SONAR REPORTS SHOT WAS ";
510 IF Y>B THEN PRINT "NORTH";
520 IF Y<B THEN PRINT "SOUTH";
530 IF X>A THEN PRINT "EAST";
540 IF X<A THEN PRINT "WEST";
550 IF Y<>B OR X<>A THEN PRINT " AND";
560 IF Z>C THEN PRINT " TOO LOW."
570 IF Z<C THEN PRINT " TOO HIGH."
580 IF Z=C THEN PRINT " DEPTH OK."
590 RETURN
600 END

```